PUBLISHED ELEC. AT
www.maplesoft.com

A PROCEDURE FOR FINDING THE k^{TH} POWER OF A MATRIX

Branko Malešević
Faculty of Electrical Engineering,
University of Belgrade, Serbia
malesh@EUnet.yu

Ivana Jovović
PhD student, Faculty of Mathematics,
University of Belgrade, Serbia
ivana121@EUnet.yu

1 Introduction

This worksheet demonstrates the use of Maple in Linear Algebra.

We give a new procedure (`PowerMatrix`) in Maple for finding the k^{th} power of n-by-n square matrix `A`, in a symbolic form, for any positive integer k, $k \geq n$. The algorithm is based on an application of Cayley-Hamilton theorem. We used the fact that the entries of the matrix $\mathtt{A}^k$ satisfy the same recurrence relation which is determined by the characteristic polynomial of the matrix `A` (see [**1**]). The order of these recurrences is $n - d$, where d is the lowest degree of the characteristic polynomial of the matrix `A`.

For non-singular matrices the procedure can be extended for k not only a positive integer.

2 Initialization

```
> restart:
  with(LinearAlgebra):
```

2.1 Procedure Definition

2.1.1 PowerMatrix

Input data are a square matrix `A` and a parameter k. Elements of the matrix `A` can be numbers and/or parameters. The parameter k can take numeric value or be a symbol. The output data is the k^{th} power of the matrix. The procedure `PowerMatrix` is as powerful as the procedure `rsolve`.

```
> PowerMatrix := proc(A::Matrix,k)
  local i,j,m,r,q,n,d,f,P,F,C;
  P := x->CharacteristicPolynomial(A,x);
  n := degree(P(x),x);
  d := ldegree(P(x),x);
```

```
F:= (i,j)->rsolve(sum(coeff(P(x),x,m)*f(m+q),m=0..n)=0,seq(f(r)=(A^r)[i,j],
r=d+1..n),f);
C:= q->Matrix(n,n,F);
if (type(k,integer)) then return(simplify(A^k)) elif (Determinant(A)=0 and
not type(k,numeric)) then printf("The %a^th power of the matrix for %a>=%d:",
k,k,n) elif (Determinant(A)=0 and type(k,numeric)) then return(simplify(A^k)) fi;
return(simplify(subs(q=k,C(q))));
end:
```

3 Examples

3.1 Example 1.

> A:= Matrix([[4,-2,2],[-5,7,-5],[-6,6,-4]]);

$$A := \begin{bmatrix} 4 & -2 & 2 \\ -5 & 7 & -5 \\ -6 & 6 & -4 \end{bmatrix}$$

> PowerMatrix(A,k);

$$\begin{bmatrix} -2^k + 2 \cdot 3^k & 2^{(1+k)} - 2 \cdot 3^k & -2^{(1+k)} + 2 \cdot 3^k \\ -5 \cdot 3^k + 5 \cdot 2^k & 5 \cdot 3^k - 4 \cdot 2^k & -5 \cdot 3^k + 5 \cdot 2^k \\ 6 \cdot 2^k - 6 \cdot 3^k & -6 \cdot 2^k + 6 \cdot 3^k & -6 \cdot 3^k + 7 \cdot 2^k \end{bmatrix}$$

> Determinant(A);

$$12$$

> B:= A^(-1);

$$B := \begin{bmatrix} \frac{1}{6} & \frac{1}{3} & -\frac{1}{3} \\ \frac{5}{6} & -\frac{1}{3} & \frac{5}{6} \\ 1 & -1 & \frac{3}{2} \end{bmatrix}$$

> PowerMatrix(B,k);

$$\begin{bmatrix} -2^{(-k)} + 2 \cdot 3^{(-k)} & 2^{(1-k)} - 2 \cdot 3^{(-k)} & -2^{(1-k)} + 2 \cdot 3^{(-k)} \\ -5 \cdot 3^{(-k)} + 5 \cdot 2^{(-k)} & 5 \cdot 3^{(-k)} - 4 \cdot 2^{(-k)} & -5 \cdot 3^{(-k)} + 5 \cdot 2^{(-k)} \\ -6 \cdot 3^{(-k)} + 6 \cdot 2^{(-k)} & -6 \cdot 2^{(-k)} + 6 \cdot 3^{(-k)} & -6 \cdot 3^{(-k)} + 7 \cdot 2^{(-k)} \end{bmatrix}$$

3.2 Example 2.

> A:= Matrix([[1-p,p],[p,1-p]]);

$$A := \begin{bmatrix} 1-p & p \\ p & 1-p \end{bmatrix}$$

> PowerMatrix(A,k);

$$\begin{bmatrix} \frac{(1-2\,p)^k}{2}+\frac{1}{2} & -\frac{(1-2\,p)^k}{2}+\frac{1}{2} \\ -\frac{(1-2\,p)^k}{2}+\frac{1}{2} & \frac{(1-2\,p)^k}{2}+\frac{1}{2} \end{bmatrix}$$

The example is from [**4**], page 272, exercise 19.

3.3 Example 3.

> A := Matrix([[a,b,c],[d,e,f],[g,h,i]]);

$$\mathtt{A} := \begin{bmatrix} a & b & c \\ d & e & f \\ g & h & i \end{bmatrix}$$

> PowerMatrix(A,k)[1,1];

$$\sum\nolimits_{\_R = \mathtt{RootOf}\left((gbf+hdc+iea-gce-hfa-idb)\_Z^3+(gc+hf+db-ie-ia-ea)\_Z^2+(i+e+a)\_Z-1 \right)}$$
$$\Big[\left(\_R^2 ie-\_R^2 hf-\_Re-\_Ri+1\right)\left(\frac{1}{\_R}\right)^k \Big/ \Big((3\_R^2 gbf+3\_R^2 hdc+3\_R^2 iea$$
$$-3\_R^2 gce-3\_R^2 hfa-3\_R^2 idb+2\_Rgc+2\_Rhf+2\_Rdb-2\_Rie-2\_Ria-2\_Rea+\ i+e+a)\_R\Big)\Big]$$

\# Warning!
In this example MatrixPower and MatrixFuction procedures cannot be done in real-time.

\# MatrixPower(A,k)[1,1];

\# MatrixFunction(A,v^k,v)[1,1];

3.4 Example 4.

> A := Matrix([[0,0,1,0,1],[1,0,0,0,1],[0,0,0,1,1],[0,1,0,0,1],[1,1,1,1,0]]);

$$\mathtt{A} := \begin{bmatrix} 0 & 0 & 1 & 0 & 1 \\ 1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & 0 & 1 \\ 1 & 1 & 1 & 1 & 0 \end{bmatrix}$$

> PowerMatrix(A,k)[1,5];

$$-\frac{\sqrt{17}\left(\frac{1}{2}-\frac{\sqrt{17}}{2}\right)^k}{17}+\frac{\sqrt{17}\left(\frac{1}{2}+\frac{\sqrt{17}}{2}\right)^k}{17}$$

Replace ':' with ';' and see result!

> MatrixPower(A,k)[1,5]:

> assume(m::integer):simplify(MatrixPower(A,k)[1,5]):

The example is from [**3**], page 101.

3.5 Example 5. and Example 6.

Pay attention what happens for singular matrices.

3.5.1 Example 5.

```
> A := Matrix([[0,2,1,3],[0,0,-2,4],[0,0,0,5],[0,0,0,0]]);
```

$$\mathtt{A} := \begin{bmatrix} 0 & 2 & 1 & 3 \\ 0 & 0 & -2 & 4 \\ 0 & 0 & 0 & 5 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> PowerMatrix(A,2);
```

$$\begin{bmatrix} 0 & 0 & -4 & 13 \\ 0 & 0 & 0 & -10 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> PowerMatrix(A,3);
```

$$\begin{bmatrix} 0 & 0 & 0 & -20 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> PowerMatrix(A,k);
```

The k^{th} power of the matrix A for $k \geq 4$:

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> MatrixPower(A,k);
```

Error, (in LinearAlgebra:-LA_Main:-MatrixPower)
power k is not defined for this Matrix

```
> MatrixFunction(A,v^k,v);
```

Error, (in LinearAlgebra:-LA_Main:-MatrixFunction)
Matrix function $\mathtt{v}^k$ is not defined for this Matrix

The example is from [**2**], page 151, exercise 23.

3.5.2 Example 6.

```
> A := Matrix([[1,1,1,0],[1,1,1,-1],[0,0,-1,1],[0,0,1,-1]]);
```

$$\mathtt{A} := \begin{bmatrix} 1 & 1 & 1 & 0 \\ 1 & 1 & 1 & -1 \\ 0 & 0 & -1 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix}$$

```
> PowerMatrix(A,k);
> The k^th power of the matrix for k ≥ 4:
```

$$\begin{bmatrix} 2^{(-1+k)} & 2^{(-1+k)} & \frac{(-1)^{(1+k)}\cdot 2^k}{16}+\frac{5\cdot 2^k}{16} & -\frac{2^k}{16}+\frac{(-1)^k\cdot 2^k}{16} \\ 2^{(-1+k)} & 2^{(-1+k)} & \frac{5\cdot 2^k}{16}+\frac{5\cdot(-1)^{(1+k)}\cdot 2^k}{16} & \frac{5\cdot(-1)^k\cdot 2^k}{16}-\frac{2^k}{16} \\ 0 & 0 & (-1)^k\cdot 2^{(-1+k)} & (-1)^{(1+k)}\cdot 2^{(-1+k)} \\ 0 & 0 & (-1)^{(1+k)}\cdot 2^{(-1+k)} & (-1)^k\cdot 2^{(-1+k)} \end{bmatrix}$$

```
> MatrixPower(A,k);
  Error, (in LinearAlgebra:-LA_Main:-MatrixPower)
  power k is not defined for this Matrix

> MatrixFunction(A,v^k,v);
  Error, (in LinearAlgebra:-LA_Main:-MatrixFunction)
  Matrix function v^k is not defined for this Matrix
```

4 References

[1] Branko Malešević: *Some combinatorial aspects of the composition of a set of functions*, NSJOM 2006 (**36**), 3-9, URLs: http://www.im.ns.ac.yu/NSJOM/Papers/36_1/NSJOM_36_1_003_009.pdf, http://arxiv.org/abs/math.CO/0409287.

[2] John B. Johnston, G. Baley Price, Fred S. Van Vleck: *Linear Equations and Matrices*, Addison-Wesley, 1966.

[3] Carl D. Meyer: *Matrix Analysis and Applied Linear Algebra Book and Solutions Manual* SIAM, 2001.

[4] Robert Messer: *Linear Algebra Gateway to Mathematics*, New York, Harper-Collins College Publisher, 1993.

5 Conclusions

This procedure has an educational character. It is an interesting demonstration for finding the k^{th} power of a matrix in a symbolic form. Sometimes, it gives solutions in the better form than the existing procedure `MatrixPower` (see example 4.). See also example 5. and example 6., where we consider singular matrices. In these cases the procedure `MatrixPower` does not give a solution. The procedure `PowerMatrix` calculates the k^{th} power of any singular matrices. In some examples it is possible to get a solution in the better form with using the procedure `allvalues` (see example 3.).